

# Wahrscheinlichkeitsrechnung

VON JOHANN WEILHARTER

# Inhaltsverzeichnis

Inhaltsverzeichnis .....	2
--------------------------	---

<b>Mengen</b> .....	4
<b>Kombinatorik</b> .....	6
Abzählprinzip .....	6
Permutationen .....	8
Kombinationen .....	10
<b>Beschreibende Statistik</b> .....	10
Kennzahlen .....	10
<b>Regressionsrechnung</b> .....	16
Lineare Regression .....	16
Quadratische Regression .....	17
<b>Wahrscheinlichkeitsrechnung</b> .....	17
Zufallsexperimente .....	17
Mengenlehre .....	18
Laplace Wahrscheinlichkeit .....	20
Der Additionssatz .....	25
<b>Wahrscheinlichkeitsverteilungen</b> .....	26
Binomialverteilung .....	26
Poissonverteilung .....	30
Normalverteilung .....	31
<b>Tabellenverzeichnis</b> .....	33
<b>Abbildungsverzeichnis</b> .....	34


# Mengen

In der Stochastik ist es sinnvoll, die Schreibweise der Mengenlehre zu verwenden. Die zugrunde liegende Boolesche Algebra beschreibt auch dieses Fachgebiet.

Eine Menge ist eine Zusammenfassung von wohlunterschiedenen Objekten unserer Anschauung oder unseres Denkens zu einem Ganzen.

**Beispiel 1.** Gegeben seien 3 Mengen A,B,C. Man berechne

- a)  $A \cap B$ ,
- b)  $A \cup B$ ,
- c)  $A \cap (B \cup C)$ ,
- d)  $A \cup (B \cap C)$ ,
- e)  $A \cup B \cup C$ ,
- f)  $A \cap B \cap C$



**Abbildung 1.** Durchschnitt von Mengen

Der Durchschnitt von 2 Mengen A und B ist  $A \cap B = \{x | x \in A \wedge x \in B\}$

```
(%i1) A: {1,2,3,4,5};B: {4,5,6,7};C: {6,7,8}
```

```
(%o1) {1, 2, 3, 4, 5}
```

```
(%o2) {4, 5, 6, 7}
```

```
(%o3) {6, 7, 8}
```

```
(%i4) intersection(A,B)
```

```
(%o4) {4, 5}
```

```
(%i5) intersect(A,B)
```

```
(%o5) {4, 5}
```

```
(%i6) union(A,B)
```

```

(%o6) {1, 2, 3, 4, 5, 6, 7}
(%i7) intersection(A, union(B, C))
(%o7) {4, 5}
(%i8) union(A, intersection(B, C))
(%o8) {1, 2, 3, 4, 5, 6, 7}
(%i9) union(A, B, C)
(%o9) {1, 2, 3, 4, 5, 6, 7, 8}
(%i10) intersection(A, B, C)
(%o11) {}
(%i12)

```




Abbildung 2. Vereinigung von Mengen

Die Vereinigung zweier Mengen ist  $A \cup B = \{x | x \in A \vee x \in B\}$

Durchschnitt	$A \cap B$	intersect(A,B)
		intersection(A,B)
Vereinigung	$A \cup B$	union(A,B)

Tabelle 1. Durchschnitt und Vereinigung von Mengen

**Aufgabe 1.**  $A = \{1, 2, 3, 4, 5, 6\}$ ,  $B = \{3, 4, 5, 6, 7\}$ . Man berechne

- $A \cap B$
- $A \cup B$
- $B \cup (A \cap B)$
- $A \cap (B \cup C)$

**Aufgabe 2.**  $A = \{a, b, c, d\}$ ,  $B = \{c, d, e\}$ . Mnn berechne

- $A \cup B$
- $A \cap B$
- $A \cap (B \cup C)$

d)  $A \cup (B \cap C)$

**Beispiel 2.** Man erstelle eine Funktion zum Erzeugen von Mengen mit  $k$  Elementen.

```
(%i1) f(k):=makeset(a[i],[i],makelist([i],i,1,k))
(%o1) f(k):= makeset(a_i,[i],makelist([i],i,1,k))
(%i2) f(0)
(%o2) {}
(%i3) f(1)
(%o3) {a_1}
(%i4) f(2)
(%o4) {a_1,a_2}
(%i5) f(3)
(%o5) {a_1,a_2,a_3}
(%i6) f(4)
(%o6) {a_1,a_2,a_3,a_4}
(%i7) f(5)
(%o7) {a_1,a_2,a_3,a_4,a_5}
(%i8)
```

Maxima	Erklärung
makeset	erzeugt eine Menge
makelist	erzeugt eine Liste

**Tabelle 2.** Erzeugung von Mengen und Listen

## Kombinatorik

### Abzählprinzip

**Beispiel 3.** Jemand hat 3 Krawatten und 4 Anzüge. Auf wie viele Arten kann er sich anziehen?

Lösung:  $3 \times 4 = 12$

```
(%i1) K:{k1,k2,k3}
(%o1) {k1,k2,k3}
(%i2) n1:cardinality(K)
(%o2) 3
(%i3) A:{a1,a2,a3,a4}
```

```

(%o3) {a1, a2, a3, a4}
(%i4) n2:cardinality(A)
(%o4) 4
(%i5) M1:cartesian_product(A,K)
(%o5) {[a1, k1], [a1, k2], [a1, k3], [a2, k1], [a2, k2], [a2, k3], [a3, k1], [a3, k2], [a3, k3], [a4, k1], [a4, k2], [a4, k3]}
(%i6) M2:cartesian_product(K,A)
(%o6) {[k1, a1], [k1, a2], [k1, a3], [k1, a4], [k2, a1], [k2, a2], [k2, a3], [k2, a4], [k3, a1], [k3, a2], [k3, a3], [k3, a4]}
(%i7) m1:cardinality(M1)
(%o7) 12
(%i8) m2:cardinality(M2)
(%o8) 12
(%i9) is(m1=m2)
(%o9) true
(%i10) is(m1=n1*n2)
(%o10) true
(%i11)

```

**Aufgabe 3.** Jemand hat 5 Paar Schuhe, 3 Anzüge und 4 Krawatten. Wie viele Kombinationen sind möglich?

**Beispiel 4.** Bei einer Party sind 5 Burschen und 4 Mädchen. Wie viele Tanzpaare können gebildet werden?

Lösung:  $5 \times 4 = 20$

```

(%i11) B:{b1,b2,b3,b4,b5}$
(%i12) M:{m1,m2,m3,m4}$
(%i13) TP:cartesian_product(B,M)
(%o14) {[b1, 12], [b1, m3], [b1, m4], [b2, 12], [b2, m3], [b2, m4], [b3, 12], [b3, m3], [b3, m4], [b4, 12], [b4, m3], [b4, m4], [b5, 12], [b5, m3], [b5, m4]}
(%i15) is(cardinality(B)*cardinality(M)=cardinality(TP))
(%o15) true
(%i16)

```

**Aufgabe 4.** Man berechne die Produktmenge  $A \times B$  für  $A=\{1,2,3\}$  und  $B=\{a,b,c,d\}$

Mächtigkeit	cardinality	Anzahl der Elemente einer Menge
Produktmenge	cartesian_product	$A \times B = \{(a, b)   x \in A \wedge y \in B\}$

**Tabelle 3.** Mächtigkeit und Produktmenge

**Beispiel 5.** Wenn jemand 1 Hut, 2 Anzüge und 3 Krawatten hat, gibt es 6 Kombinationsmöglichkeiten.




Abbildung 3. Fundamentales Abzählprinzip

## Permutationen

(%i1) A0:{\$}

(%i2) A1:{a1}\$

(%i3) A2:{a1,a2}\$

(%i4) A3:{a1,a2,a3}\$

(%i5) A4:{a1,a2,a3,a4}\$

(%i6) A5:{a1,a2,a3,a4,a5}\$

(%i7) P0:permutations(A0)

(%o8) {[]}

(%i9) P1:permutations(A1)

(%o9) permutations({a1})

(%i10) P2:permutations(A2)

(%o10) {[a1, a2], [a2, a1]}

(%i11) P3:permutations(A3)

(%o11) {[a1, a2, a3], [a1, a3, a2], [a2, a1, a3], [a2, a3, a1], [a3, a1, a2], [a3, a2, a1]}

(%i12) P4:permutations(A4)

(%o12) {[a1, a2, a3, a4], [a1, a2, a4, a3], [a1, a3, a2, a4], [a1, a3, a4, a2], [a1, a4, a2, a3], [a1, a4, a3, a2], [a2, a1, a3, a4], [a2, a1, a4, a3], [a2, a3, a1, a4], [a2, a3, a4, a1], [a2, a4, a1, a3], [a2, a4, a3, a1], [a3, a1, a2, a4], [a3, a1, a4, a2], [a3, a2, a1, a4], [a3, a2, a4, a1], [a3, a4, a1, a2], [a3, a4, a2, a1], [a4, a1, a2, a3], [a4, a1, a3, a2], [a4, a2, a1, a3], [a4, a2, a3, a1], [a4, a3, a1, a2], [a4, a3, a2, a1]}

```

(%i13) P5:=permutations(A5)$
(%i23) n0:=length(P0)
(%o16) 1
(%i17) n1:=length(P1)
(%o17) 1
(%i18) n2:=length(P2)
(%o18) 2
(%i19) n3:=length(P3)
(%o19) 6
(%i20) n4:=length(P4)
(%o20) 24
(%i21) n5:=length(P5)
(%o21) 120
(%i22)

```

Da Permutationen Listen sind, muss man *length* verwenden. Bei Mengen verwendet man *cardinality*.

**Beispiel 6.** Die Mächtigkeit der Menge der Permutationen ist zu bestimmen.

```

(%i1) f(k):=makeset(a[i],[i],makelist([i],i,1,k))
(%o1) f(k):=makeset(a_i,[i],makelist([i],i,1,k))
(%i2) perm(k):=permutations(f(k))
(%o2) perm(k):=permutations(f(k))
(%i3) n(k):=cardinality(perm(k))
(%o3) n(k):=cardinality(perm(k))
(%i4) makelist([k,n(k)],k,0,5)
(%o5) [[0, 1], [1, 1], [2, 2], [3, 6], [4, 24], [5, 120]]
(%i6)

```

**Aufgabe 5.** Erzeuge mit

```
f(k):=makeset(a[i],[i],makelist([i],i,1,k))
```

eine Menge mit 10 Elementen.

**Aufgabe 6.** Erzeuge mit

```
f(k):=makeset(a[i],[i],makelist([i],i,1,k))
f(k):=makeset(a_i,[i],makelist([i],i,1,k))
perm(k):=permutations(f(k))
```

die Permutationen einer Menge mit 4 Elementen.

**Beispiel 7.** n-Fakultät (die Anzahl der Permutationen mit n Elementen) soll mit einer stückweise definierten Funktion berechnet werden.

```

(%i1) fakultaet(x):=if x=0 then 1 else prod(i,i,1,x)
(%o1) fakultaet(x):= if x = 0 then 1 else product(i, i, 1, x)
(%i2) fakultaet(5)
(%o2) 120
(%i3) fakultaet(0)
(%o3) 1
(%i4) fakultaet(1)
(%o4) 1
(%i5) fakultaet(7)
(%o5) 5040
(%i6)

```

## Kombinationen

## Beschreibende Statistik

### Kennzahlen

**Beispiel 8.** Gegeben ist die Urliste einer statistischen Erhebung. Man berechne

- a) den Mittelwert
- b) die Varianz
- c) die Streuung
- d) das Minimum
- e) das Maximum
- f) die Bandbreite
- g) den Median

```

(%i1) load(descriptive)$
(%i2) x: [1,5,3,4,6,7,2,3,4,5,6,3,8,3,4,5,2]
(%o2) [1, 5, 3, 4, 6, 7, 2, 3, 4, 5, 6, 3, 8, 3, 4, 5, 2]
(%i3) Mittelwert:mean(x)
(%o3)  $\frac{71}{17}$ 
(%i4) Varianz:var(x)
(%o4)  $\frac{960}{289}$ 

```

```

(%i5) Streuung:std(x)
(%o5)  $\frac{8\sqrt{15}}{17}$ 
(%i6) Minimum:mini(x)
(%o6) 1
(%i7) Maximum:maxi(x)
(%o7) 8
(%i8) Bandbreite:range(x)
(%o8) 7
(%i9) Median:median(x)
(%o9) 4
(%i10)

```

**Beispiel 9.** Das Maximum und das Minimum einer Liste sind zu bestimmen. Dazu soll das Unterprogramm „descriptive“ verwendet werden.

```

(%i18) load(descriptive)$
(%i19) liste:[1,4,3,2,5,3,6,4,7,3,2]
(%o19) [1, 4, 3, 2, 5, 3, 6, 4, 7, 3, 2]
(%i20) minimum:mini(liste)
(%o20) 1
(%i21) maximum:maxi(liste)
(%o21) 7
(%i22)

```

**Beispiel 10.** Eine Schularbeit brachte folgendes Ergebnis: 1,2,1,3,2,4,4,3,2,4,3,5,1,2,4,3,2,1,4,4,5 . Man berechne

- die gruppierte Liste,
- den Erwartungswert,
- die Varianz und
- die Streuung.

```

(%i12) noten:[1,2,1,3,2,4,3,2,4,3,5,1,2,4,3,2,1,4,4,5]
(%o12) [1, 2, 1, 3, 2, 4, 3, 2, 4, 3, 5, 1, 2, 4, 3, 2, 1, 4, 4, 5]
(%i13) load(descriptive)$
(%i14) gruppierung:discrete_freq(noten)
(%o15) [[1, 2, 3, 4, 5], [4, 5, 4, 5, 2]]
(%i16) X:gruppierung[1]
(%o16) [1, 2, 3, 4, 5]

```

```

(%i17) H:gruppierung[2]
(%o17) [4, 5, 4, 5, 2]
(%i18) n:length(X)
(%o18) 5
(%i19) N:sum(H[i],i,1,n)
(%o19) 20
(%i20) P:H/N
(%o20)  $\left[ \frac{1}{5}, \frac{1}{4}, \frac{1}{5}, \frac{1}{4}, \frac{1}{10} \right]$ 
(%i21) m:sum(P[i]*X[i],i,1,n)
(%o21)  $\frac{14}{5}$ 
(%i22) v:sum(P[i]*(X[i]-m)**2,i,1,n)
(%o22)  $\frac{83}{50}$ 
(%i23) s:sqrt(v)
(%o23)  $\frac{\sqrt{83}}{5\sqrt{2}}$ 
(%i24)

```

**Beispiel 11.** An einer vielbefahrenen Kreuzung gibt es maximal 4 Verkehrsunfälle pro Woche. Ungefähr ein Jahr lang werden die Verkehrsunfälle beobachtet.

```

(%i4) load(vect)$
(%i5) x:[0,1,2,3,4]
(%o5) [0, 1, 2, 3, 4]
(%i6) h:[30,10,8,3,1]
(%o6) [30, 10, 8, 3, 1]
(%i7) n:length(x)
(%o7) 5
(%i8) s:makelist(1,i,1,n)
(%o8) [1, 1, 1, 1, 1]
(%i9) N:h.s
(%o9) 52
(%i10) p:h/N
(%o10)  $\left[ \frac{15}{26}, \frac{5}{26}, \frac{2}{13}, \frac{3}{52}, \frac{1}{52} \right]$ 
(%i11) E:p.x
(%o11)  $\frac{3}{4}$ 

```

```

(%i12) V:p. (x-E)**2
(%o12)  $\frac{223}{208}$ 
(%i13) S:sqrt(V)
(%o13)  $\frac{\sqrt{223}}{4\sqrt{13}}$ 
(%i14) E:E, numer
(%o14) 0.75
(%i15) E:floor(E*1000+0.5)/1000.0
(%o15) 0.75
(%i16) print("Erwartungswert = ",E)$
Erwartungswert = 0.75
(%i18)

```

**Beispiel 12.** Für eine gegebene Liste sollen

- a) das arithmetische Mittel
  - b) das harmonische Mittel
  - c) das geometrische Mittel
- bestimmt werden.

```

(%i9) load(descriptive)$
(%i10) x:[1,2,3,4,5]
(%o10) [1, 2, 3, 4, 5]
(%i11) m:mean(x)
(%o11) 3
(%i12) hm:harmonic_mean(x)
(%o12)  $\frac{300}{137}$ 
(%i13) gm:geometric_mean(x)
(%o13)  $120^{\frac{1}{5}}$ 
(%i14) m:m, numer; m:floor(m*1000+0.5)/1000.0
(%o15) 3
(%o16) 3.0
(%i17) hm:hm, numer; hm:floor(hm*1000+0.5)/1000.0
(%o17) 2.18978102189781
(%o18) 2.19
(%i19) gm:gm, numer; gm:floor(gm*1000+0.5)/1000.0
(%o19) 2.605171084697352
(%o20) 2.605

```

(%i21)

**Beispiel 13.** Gegeben sei eine Häufigkeitsverteilung. Man berechne

- a) den Erwartungswert  $E(X)$
- b)  $E(3X+4)$
- c)  $3E(X)+4$
- d)  $E(X-2)$
- e)  $E(X-2)$

(%i99) `x:[1,2,3,4,5]`

(%o99) `[1,2,3,4,5]`

(%i100) `h:[4,2,10,3,6]`

(%o100) `[4,2,10,3,6]`

(%i101) `n:length(x)`

(%o101) `5`

(%i102) `N:sum(h[i],i,1,n)`

(%o102) `25`

(%i103) `p:h/N`

(%o103) `[ 4/25, 2/25, 2/5, 3/25, 6/25 ]`

(%i104) `m:sum(p[i]*x[i],i,1,n)`

(%o104) `16/5`

(%i105) `m1:sum(p[i]*(3*x[i]+4),i,1,n)`

(%o105) `68/5`

(%i106) `is(m1=3*m+4)`

(%o106) `true`

(%i107) `m2:sum(p[i]*(x[i]-2),i,1,n)`

(%o107) `6/5`

(%i108) `is(m2=m-2)`

(%o108) `true`

(%i109)

**Beispiel 14.** In einer Häufigkeitsverteilung ist ein Wert unbekannt, z.B.  $h_1$  (es könnte auch  $x_4$  usw. sein). Wenn der Erwartungswert bekannt ist, lässt sich dieser fehlende Wert bestimmen.

In der Musteraufgabe ist der Erwartungswert  $m = 3$

```

(%i20) x:[1,2,3,4,5]
(%o20) [1, 2, 3, 4, 5]
(%i21) h:[h1,2,10,2,1]
(%o21) [h1, 2, 10, 2, 1]
(%i22) m:3
(%o22) 3
(%i23) n:length(x)
(%o23) 5
(%i24) N:sum(h[i],i,1,n)
(%o24) h1 + 15
(%i25) p:h/N
(%o25)  $\left[ \frac{h1}{h1 + 15}, \frac{2}{h1 + 15}, \frac{10}{h1 + 15}, \frac{2}{h1 + 15}, \frac{1}{h1 + 15} \right]$ 
(%i26) g:m=sum(p[i]*x[i],i,1,n)
(%o26)  $3 = \frac{h1}{h1 + 15} + \frac{47}{h1 + 15}$ 
(%i27) l:solve(g,h1)
(%o27) [h1 = 1]
(%i28)

```

**Beispiel 15.** In einer Häufigkeitsverteilung ist ein Wert unbekannt, z.B.  $x_1$  (es könnte auch  $h_4$  usw. sein). Wenn die Varianz bekannt ist, lässt sich dieser fehlende Wert bestimmen.

In der Musteraufgabe ist die Varianz  $v = 0,75$

```

(%i10) x:[x1,2,3,4,5]
(%o10) [x1, 2, 3, 4, 5]
(%i11) h:[1,2,10,2,1]
(%o11) [1, 2, 10, 2, 1]
(%i12) v:0.75
(%o12) 0.75
(%i13) n:length(x)
(%o13) 5
(%i14) N:sum(h[i],i,1,n)
(%o14) 16
(%i15) p:h/N
(%o15)  $\left[ \frac{1}{16}, \frac{1}{8}, \frac{5}{8}, \frac{1}{8}, \frac{1}{16} \right]$ 
(%i16) m:sum(p[i]*x[i],i,1,n)
(%o16)  $\frac{x1}{16} + \frac{47}{16}$ 

```

```
(%i17) g:sum(p[i]*(x[i]-m)**2,i,1,n)=v
```

```
(%o18) 
$$\frac{\left(\frac{15x_1}{16} - \frac{47}{16}\right)^2}{16} + \frac{\left(\frac{33}{16} - \frac{x_1}{16}\right)^2}{16} + \frac{\left(\frac{17}{16} - \frac{x_1}{16}\right)^2}{8} + \frac{5\left(\frac{1}{16} - \frac{x_1}{16}\right)^2}{8} + \frac{\left(-\frac{x_1}{16} - \frac{15}{16}\right)^2}{8} = 0.75$$

```

```
(%i19) l:solve(g,x1)
```

'rat' replaced -0.75 by -3//4 = -0.75

```
(%o19) 
$$\left[ x_1 = 1, x_1 = \frac{79}{15} \right]$$

```

```
(%i20)
```

## Regressionsrechnung

### Lineare Regression

**Beispiel 16.** Man bestimme die lineare Funktion, die zur folgenden Wertetabelle passt:

x	1	2	3	4	5
y	11	21	31	41	51

```
(%i1) x:[1,2,3,4,5]
```

```
(%o1) [1, 2, 3, 4, 5]
```

```
(%i2) y:[11,21,31,41,51]
```

```
(%o2) [11, 21, 31, 41, 51]
```

```
(%i3) n:length(x)
```

```
(%o3) 5
```

```
(%i4) sx2:sum(x[i]**2,i,1,n)
```

```
(%o4) 55
```

```
(%i5) sx:sum(x[i],i,1,n)
```

```
(%o5) 15
```

```
(%i6) sxy:sum(x[i]*y[i],i,1,n)
```

```
(%o6) 565
```

```
(%i7) sy:sum(y[i],i,1,n)
```

```
(%o7) 155
```

```
(%i8) g1:a*sx2+b*sx=sxy
```

```
(%o8)  $15b + 55a = 565$ 
```

```
(%i9) g2:a*sx+b*n=sy
```

```
(%o9)  $5b + 15a = 155$ 
```

```
(%i10) l:solve([g1,g2],[a,b])
```

```
(%o10) [[a = 10, b = 1]]
```

```
(%i11) gerade:Y=a*X+b,l
```

(%o11)  $Y = 10X + 1$

(%i12)

**Aufgabe 7.** Erstelle die Regressionsgeraden zu den folgenden Wertetabellen

a) 

x	1	2	3	4	5
y	9	19	29	39	49

b) 

x	10	20	30	40	50	60	70
y	105	115	125	135	145	155	165

## Quadratische Regression

## Wahrscheinlichkeitsrechnung

### Zufallsexperimente

**Beispiel 17.** Das zwölfmalige Werfen eines Würfels soll mit dem Zufallsgenerator simuliert werden.

a) mit FOR-Schleife

b) mit makelist

```
(%i22) for i:1 thru 12 do print(random(6)+1)
```

```
3  
1  
3  
6  
5  
2  
6  
6  
1  
2  
2  
6
```

```
(%o22) done
```

```
(%i23) makelist(random(6)+1,k,1,12)
```

```
(%o24) [1, 6, 1, 4, 5, 2, 5, 6, 2, 6, 2, 1]
```

```
(%i25)
```

**Beispiel 18.** Man bestimme den Ergebnisraum bei den folgenden Zufallsexperimenten:

a) zweimaliger Münzwurf,

b) zuerst Münzwurf, dann Würfeln,

c) zweimaliges Werfen eines Würfels.

```
(%i47) M1: {K, Z}
```

```

(%o48) {K, Z}
(%i49) M2:{K,Z}
(%o49) {K, Z}
(%i50) S:cartesian_product(M1,M2)
(%o50) {[K, K], [K, Z], [Z, K], [Z, Z]}
(%i51) M:{K,Z}
(%o51) {K, Z}
(%i52) W:{1,2,3,4,5,6}
(%o52) {1, 2, 3, 4, 5, 6}
(%i53) S:cartesian_product(M,W)
(%o53) {[K, 1], [K, 2], [K, 3], [K, 4], [K, 5], [K, 6], [Z, 1], [Z, 2], [Z, 3], [Z, 4], [Z, 5], [Z, 6]}
(%i54) W1:{1,2,3,4,5,6}
(%o54) {1, 2, 3, 4, 5, 6}
(%i55) W2:{1,2,3,4,5,6}
(%o55) {1, 2, 3, 4, 5, 6}
(%i56) S:cartesian_product(W1,W2)
(%o56) {[1, 1], [1, 2], [1, 3], [1, 4], [1, 5], [1, 6], [2, 1], [2, 2], [2, 3], [2, 4], [2, 5], [2, 6], [3, 1], [3, 2], [3, 3], [3, 4], [3, 5], [3, 6], [4, 1], [4, 2], [4, 3], [4, 4], [4, 5], [4, 6], [5, 1], [5, 2], [5, 3], [5, 4], [5, 5], [5, 6], [6, 1], [6, 2], [6, 3], [6, 4], [6, 5], [6, 6]}
(%i57)

```

## Mengenlehre

Wie schon erwähnt, ist es zweckmäßig, die *Boolesche Algebra*, die hinter der Mengenlehre steckt, auch in der Wahrscheinlichkeitsrechnung zu verwenden.

**Beispiel 19.** Gegeben sind zwei Mengen A und B. Man bestimme

- die Durchschnittsmenge
- die Vereinigungsmenge
- die Potenzmenge
- die Produktmenge
- die Mächtigkeit der Potenzmenge
- die Mächtigkeit der Produktmenge
- die Permutationen
- die Mächtigkeit der Permutationen

```
(%i57) A:{1,2,3}
```

```
(%o57) {1, 2, 3}
```

```

(%i58) B:{2,3,4}
(%o58) {2,3,4}
(%i59) Durchschnitt:intersection(A,B)
(%o59) {2,3}
(%i60) Vereinigung:union(A,B)
(%o60) {1,2,3,4}
(%i61) Potenzmenge_A:powerset(A)
(%o61) {{},{1},{1,2},{1,2,3},{1,3},{2},{2,3},{3}}
(%i62) Potenzmenge_B:powerset(B)
(%o62) {{},{2},{2,3},{2,3,4},{2,4},{3},{3,4},{4}}
(%i63) Produktmenge_A_B:cartesian_product(A,B)
(%o63) {[1,2],[1,3],[1,4],[2,2],[2,3],[2,4],[3,2],[3,3],[3,4]}
(%i64) Produktmenge_B_A:cartesian_product(B,A)
(%o64) {[2,1],[2,2],[2,3],[3,1],[3,2],[3,3],[4,1],[4,2],[4,3]}
(%i65) n_A:cardinality(A)
(%o65) 3
(%i66) n_Potenzmenge_A:cardinality(Potenzmenge_A)
(%o66) 8
(%i67) is(2**n_A=n_Potenzmenge_A)
(%o67) true
(%i68) n_B:cardinality(B)
(%o68) 3
(%i69) n_Potenzmenge_B:cardinality(Potenzmenge_B)
(%o69) 8
(%i70) is(2**n_B=n_Potenzmenge_B)
(%o70) true
(%i71) n_Produktmenge_A_B:cardinality(Produktmenge_A_B)
(%o71) 9
(%i72) n_Produktmenge_B_A:cardinality(Produktmenge_B_A)
(%o72) 9
(%i73) is(n_Produktmenge_A_B=n_Produktmenge_B_A)
(%o73) true
(%i74) is(n_A*n_B=n_Produktmenge_A_B)
(%o74) true
(%i75) L_A:listify(A)
(%o75) [1,2,3]
(%i76) L_B:listify(B)
(%o76) [2,3,4]

```

```

(%i77) Perm_A:permutations(A)
(%o77) {[1, 2, 3], [1, 3, 2], [2, 1, 3], [2, 3, 1], [3, 1, 2], [3, 2, 1]}
(%i78) Perm_A:permutations(L_A)
(%o78) {[1, 2, 3], [1, 3, 2], [2, 1, 3], [2, 3, 1], [3, 1, 2], [3, 2, 1]}
(%i79) Perm_B:permutations(B)
(%o79) {[2, 3, 4], [2, 4, 3], [3, 2, 4], [3, 4, 2], [4, 2, 3], [4, 3, 2]}
(%i80) Perm_B:permutations(L_B)
(%o82) {[2, 3, 4], [2, 4, 3], [3, 2, 4], [3, 4, 2], [4, 2, 3], [4, 3, 2]}
(%i83) n_Perm_A:cardinality(Perm_A)
(%o83) 6
(%i84) n_Perm_B:cardinality(Perm_B)
(%o84) 6
(%i85) is(n_A!=n_Perm_A)
(%o85) true
(%i86) is(n_B!=n_Perm_B)
(%o86) true
(%i87)

```

## Laplace Wahrscheinlichkeit

**Beispiel 20.** Man wirft einen normalen Würfel einmal.

- Was ist das sichere Ereignis?
- Was ist das unmögliche Ereignis?
- Wie groß ist die Wahrscheinlichkeit für eine gerade Zahl?
- Wie groß ist die Wahrscheinlichkeit für eine ungerade Zahl?

```

(%i1) S:{1,2,3,4,5,6}
(%o1) {1, 2, 3, 4, 5, 6}
(%i2) U:{}
(%o2) {}
(%i3) gerade:{2,4,6}
(%o3) {2, 4, 6}
(%i4) ungerade:{1,3,5}
(%o4) {1, 3, 5}
(%i5) moeglich:cardinality(S)
(%o5) 6
(%i6) guenstig_gerade:cardinality(gerade)

```

```

(%o6) 3
(%i7) guenstig_ungerade:cardinality(ungerade)
(%o7) 3
(%i8) w_gerade:guenstig_gerade/moeglich
(%o8)  $\frac{1}{2}$ 
(%i9) w_ungerade:guenstig_ungerade/moeglich
(%o9)  $\frac{1}{2}$ 
(%i10)

```

Wahrscheinlichkeit =  $\frac{\text{Anzahl der guenstigen Faelle}}{\text{Anzahl der moeglichen Faelle}}$

**Beispiel 21.** Aus der Menge  $\{1,2,3,4,5\}$  wird zufällig eine Ziffer ausgewählt, dann noch eine. Jede Ziffer kann nur einmal ausgewählt werden. Man konstruiere einen geeigneten Ergebnisraum.

```

(%i87) G:{1,2,3,4,5}
(%o87) {1, 2, 3, 4, 5}
(%i88) P:powerset(G)
(%o88) {{}, {1}, {1, 2}, {1, 2, 3}, {1, 2, 3, 4}, {1, 2, 3, 4, 5}, {1, 2, 3, 5}, {1, 2, 4}, {1, 2, 4, 5}, {1, 2, 5}, {1, 3}, {1, 3, 4}, {1, 3, 4, 5}, {1, 3, 5}, {1, 4}, {1, 4, 5}, {1, 5}, {2}, {2, 3}, {2, 3, 4}, {2, 3, 4, 5}, {2, 3, 5}, {2, 4}, {2, 4, 5}, {2, 5}, {3}, {3, 4}, {3, 4, 5}, {3, 5}, {4}, {4, 5}, {5}}
(%i89) S:subset(P,lambda([e],is(cardinality(e)=2)))
(%o89) {{1, 2}, {1, 3}, {1, 4}, {1, 5}, {2, 3}, {2, 4}, {2, 5}, {3, 4}, {3, 5}, {4, 5}}
(%i90) m:cardinality(S)
(%o90) 10
(%i91)

```

Es gibt also 10 Möglichkeiten.

**Beispiel 22.** Ein Würfel wird zweimal geworfen (oder zwei Würfel werden einmal geworfen). Worauf setzt man besser

- auf die Ziffernsumme 9
- auf die Ziffernsumme 10 ?

```

(%i28) W:{1,2,3,4,5,6}
(%o28) {1, 2, 3, 4, 5, 6}
(%i29) S:cartesian_product(W,W)
(%o30) {[1, 1], [1, 2], [1, 3], [1, 4], [1, 5], [1, 6], [2, 1], [2, 2], [2, 3], [2, 4], [2, 5], [2, 6], [3, 1], [3, 2], [3, 3], [3, 4], [3, 5], [3, 6], [4, 1], [4, 2], [4, 3], [4, 4], [4, 5], [4, 6], [5, 1], [5, 2], [5, 3], [5, 4], [5, 5], [5, 6], [6, 1], [6, 2], [6, 3], [6, 4], [6, 5], [6, 6]}
(%i31) S1:makeset(i+j,[i,j],S)
(%o31) {2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12}

```

```

(%i32) S2:makeset(i*j,[i,j],S)
(%o32) {1, 2, 3, 4, 5, 6, 8, 9, 10, 12, 15, 16, 18, 20, 24, 25, 30, 36}
(%i33) S3:makeset(i-j,[i,j],S)
(%o33) {-5, -4, -3, -2, -1, 0, 1, 2, 3, 4, 5}
(%i34) S4:listify(S)
(%o34) [[1, 1], [1, 2], [1, 3], [1, 4], [1, 5], [1, 6], [2, 1], [2, 2], [2, 3], [2, 4], [2, 5], [2, 6], [3, 1], [3, 2], [3, 3], [3, 4], [3, 5], [3, 6], [4, 1], [4, 2], [4, 3], [4, 4], [4, 5], [4, 6], [5, 1], [5, 2], [5, 3], [5, 4], [5, 5], [5, 6], [6, 1], [6, 2], [6, 3], [6, 4], [6, 5], [6, 6]]
(%i35) n:length(S)
(%o35) 36
(%i36) S5:makelist(S4[i][1]+S4[i][2],i,1,n)
(%o36) [2, 3, 4, 5, 6, 7, 3, 4, 5, 6, 7, 8, 4, 5, 6, 7, 8, 9, 5, 6, 7, 8, 9, 10, 6, 7, 8, 9, 10, 11, 7, 8, 9, 10, 11, 12]
(%i37) S6:sort(S5)
(%o37) [2, 3, 3, 4, 4, 4, 5, 5, 5, 5, 6, 6, 6, 6, 6, 7, 7, 7, 7, 7, 7, 8, 8, 8, 8, 8, 9, 9, 9, 9, 10, 10, 10, 11, 11, 12]
(%i38) load(descriptive)
(%o38) /usr/share/maxima/5.10.0/share/contrib/descriptive/descriptive.mac
(%i39) gruppierung:discrete_freq(S6)
(%o39) [[2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12], [1, 2, 3, 4, 5, 6, 5, 4, 3, 2, 1]]
(%i40) x:gruppierung[1]
(%o40) [2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12]
(%i41) h:gruppierung[2]
(%o41) [1, 2, 3, 4, 5, 6, 5, 4, 3, 2, 1]
(%i42) n:length(x)
(%o42) 11
(%i43) N:sum(h[i],i,1,n)
(%o43) 36
(%i44) p:h/N
(%o44)  $\left[ \frac{1}{36}, \frac{1}{18}, \frac{1}{12}, \frac{1}{9}, \frac{5}{36}, \frac{1}{6}, \frac{5}{36}, \frac{1}{9}, \frac{1}{12}, \frac{1}{18}, \frac{1}{36} \right]$ 
(%i45) p[8]
(%o45)  $\frac{1}{9}$ 
(%i46) p[9]
(%o46)  $\frac{1}{12}$ 
(%i47)

```

Es gibt 4 mal die Summe 9 und nur 3 mal die Summe 10 !tve

Eine Lösungsalternative ist:

```

(%i91) W:{1,2,3,4,5,6}
(%o91) {1, 2, 3, 4, 5, 6}
(%i92) S:cartesian_product(W,W)
(%o92) {[1, 1], [1, 2], [1, 3], [1, 4], [1, 5], [1, 6], [2, 1], [2, 2], [2, 3], [2, 4], [2, 5], [2, 6], [3, 1], [3, 2], [3, 3], [3, 4], [3, 5], [3, 6], [4, 1], [4, 2], [4, 3], [4, 4], [4, 5], [4, 6], [5, 1], [5, 2], [5, 3], [5, 4], [5, 5], [5, 6], [6, 1], [6, 2], [6, 3], [6, 4], [6, 5], [6, 6]}
(%i93) G9:subset(S,lambda([e],is(e[1]+e[2]=9)))
(%o93) {[3, 6], [4, 5], [5, 4], [6, 3]}
(%i94) G10:subset(S,lambda([e],is(e[1]+e[2]=10)))
(%o94) {[4, 6], [5, 5], [6, 4]}
(%i95) W[G9]:cardinality(G9)/cardinality(S)
(%o96)  $\frac{1}{9}$ 
(%i97) W[G10]:cardinality(G10)/cardinality(S)
(%o97)  $\frac{1}{12}$ 
(%i98) is(W[G9]>W[G10])
(%o98) true
(%i99)

```

**Beispiel 23.** Wir betrachten ein Laplace-Experiment: „Werfen dreier Würfel“. Wie groß ist die Wahrscheinlichkeit dafür, dass die Augensumme mindestens 11 ist?

```

(%i1) W:{1,2,3,4,5,6}
(%o1) {1, 2, 3, 4, 5, 6}
(%i2) S:cartesian_product(W,W,W)$
(%i3) m:cardinality(S)
(%o3) 216
(%i4) G:subset(S,lambda([e],is(e[1]+e[2]+e[3]>=11)))$
(%i5) g:cardinality(G)
(%o5) 108
(%i6) W:g/m
(%o6)  $\frac{1}{2}$ 
(%i7)

```

**Beispiel 24.** Ein Würfel wird zweimal geworfen. Man betrachte die Wahrscheinlichkeiten für folgende Ereignisse:

- a) zwei gleiche Augenzahlen
- b) zwei verschiedene Augenzahlen

```

(%i7)

```

```

(%i7) W:{1,2,3,4,5,6}
(%o7) {1, 2, 3, 4, 5, 6}
(%i8) S:cartesian_product(W,W)
(%o8) {[1, 1], [1, 2], [1, 3], [1, 4], [1, 5], [1, 6], [2, 1], [2, 2], [2, 3], [2, 4], [2, 5], [2, 6], [3, 1], [3, 2], [3, 3], [3, 4], [3, 5], [3, 6], [4, 1], [4, 2], [4, 3], [4, 4], [4, 5], [4, 6], [5, 1], [5, 2], [5, 3], [5, 4], [5, 5], [5, 6], [6, 1], [6, 2], [6, 3], [6, 4], [6, 5], [6, 6]}
(%i9) G:subset(S,lambda([e],is(e[1]=e[2])))
(%o9) {[1, 1], [2, 2], [3, 3], [4, 4], [5, 5], [6, 6]}
(%i10) V:subset(S,lambda([e],is(e[1]#e[2])))
(%o10) {[1, 2], [1, 3], [1, 4], [1, 5], [1, 6], [2, 1], [2, 3], [2, 4], [2, 5], [2, 6], [3, 1], [3, 2], [3, 4], [3, 5], [3, 6], [4, 1], [4, 2], [4, 3], [4, 5], [4, 6], [5, 1], [5, 2], [5, 3], [5, 4], [5, 6], [6, 1], [6, 2], [6, 3], [6, 4], [6, 5]}
(%i11) V:setdifference(S,G)
(%o11) {[1, 2], [1, 3], [1, 4], [1, 5], [1, 6], [2, 1], [2, 3], [2, 4], [2, 5], [2, 6], [3, 1], [3, 2], [3, 4], [3, 5], [3, 6], [4, 1], [4, 2], [4, 3], [4, 5], [4, 6], [5, 1], [5, 2], [5, 3], [5, 4], [5, 6], [6, 1], [6, 2], [6, 3], [6, 4], [6, 5]}
(%i12) m:cardinality(S)
(%o12) 36
(%i13) g[G]:cardinality(G)
(%o13) 6
(%i14) g[V]:cardinality(V)
(%o14) 30
(%i15) W[G]:g[G]/m
(%o15)  $\frac{1}{6}$ 
(%i16) W[V]:g[V]/m
(%o16)  $\frac{5}{6}$ 
(%i17) W[G]+W[V]
(%o17) 1
(%i18)

```

**Beispiel 25.** Die Ziffernsumme einer dreistelligen Zahl, kann die Werte von 1 bis 27 annehmen. Warum? Man bestimme für eine gegebene Ziffernsumme aus diesem Bereich die Wahrscheinlichkeit dafür, dass eine zufällig ausgewählte dreistellige Zahl diese Ziffernsumme aufweist.

```

(%i1) summe:15
(%o1) 15
(%i2) ziffern:{0,1,2,3,4,5,6,7,8,9}
(%o2) {0, 1, 2, 3, 4, 5, 6, 7, 8, 9}
(%i3) H:ziffern;Z:ziffern;E:ziffern
(%o3) {0, 1, 2, 3, 4, 5, 6, 7, 8, 9}
(%o4) {0, 1, 2, 3, 4, 5, 6, 7, 8, 9}
(%o5) {0, 1, 2, 3, 4, 5, 6, 7, 8, 9}

```

```

(%i6) zahlen:cartesian_product(H,Z,E)$
(%i11) S:subset(zahlen,lambda([e],is(e[1]#0)))$
(%i12) m:cardinality(S)
(%o12) 900
(%i13) L:subset(S,lambda([e],is(e[1]+e[2]+e[3]=summe)))
(%o13) {[1, 5, 9], [1, 6, 8], [1, 7, 7], [1, 8, 6], [1, 9, 5], [2, 4, 9], [2, 5, 8], [2, 6, 7], [2, 7, 6], [2, 8, 5], [2, 9, 4], [3, 3, 9], [3,
4, 8], [3, 5, 7], [3, 6, 6], [3, 7, 5], [3, 8, 4], [3, 9, 3], [4, 2, 9], [4, 3, 8], [4, 4, 7], [4, 5, 6], [4, 6, 5], [4, 7, 4], [4, 8, 3], [4, 9, 2],
[5, 1, 9], [5, 2, 8], [5, 3, 7], [5, 4, 6], [5, 5, 5], [5, 6, 4], [5, 7, 3], [5, 8, 2], [5, 9, 1], [6, 0, 9], [6, 1, 8], [6, 2, 7], [6, 3, 6], [6, 4,
5], [6, 5, 4], [6, 6, 3], [6, 7, 2], [6, 8, 1], [6, 9, 0], [7, 0, 8], [7, 1, 7], [7, 2, 6], [7, 3, 5], [7, 4, 4], [7, 5, 3], [7, 6, 2], [7, 7, 1], [7,
8, 0], [8, 0, 7], [8, 1, 6], [8, 2, 5], [8, 3, 4], [8, 4, 3], [8, 5, 2], [8, 6, 1], [8, 7, 0], [9, 0, 6], [9, 1, 5], [9, 2, 4], [9, 3, 3], [9, 4, 2],
[9, 5, 1], [9, 6, 0]}
(%i14) g[L]:cardinality(L)
(%o14) 69
(%i15) W[L]:g[L]/m
(%o15)  $\frac{23}{300}$ 
(%i16) print("Die Wahrscheinlichkeit, dass die Ziffernsumme einer dreistelligen Zahl
",summe," ist, ist ",W[L])$

```

Die Wahrscheinlichkeit, dass die Ziffernsumme einer dreistelligen Zahl 15 ist, ist  $\frac{23}{300}$

```

(%i18) LS:listify(S)$
(%i20) ziffernsummen:makelist(LS[i][1]+LS[i][2]+LS[i][3],i,1,m)$
(%i21) Ziffernsummen:setify(ziffernsummen)
(%o21) {1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15, 16, 17, 18, 19, 20, 21, 22, 23, 24, 25, 26, 27}
(%i22)

```

## Der Additionssatz

$$W(A \cup B) = W(A) + W(B) - W(A \cap B)$$

**Beispiel 26.** Aus den ersten 100 Zahlen wird zufällig eine Zahl ausgewählt. Wie groß ist die Wahrscheinlichkeit, dass die ausgewählte Zahl durch 4 oder durch 6 teilbar ist.

```

(%i109) S:makelist(i,i,1,100)
(%o109) [1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15, 16, 17, 18, 19, 20, 21, 22, 23, 24, 25, 26, 27, 28, 29, 30, 31, 32,
33, 34, 35, 36, 37, 38, 39, 40, 41, 42, 43, 44, 45, 46, 47, 48, 49, 50, 51, 52, 53, 54, 55, 56, 57, 58, 59, 60, 61, 62, 63, 64,
65, 66, 67, 68, 69, 70, 71, 72, 73, 74, 75, 76, 77, 78, 79, 80, 81, 82, 83, 84, 85, 86, 87, 88, 89, 90, 91, 92, 93, 94, 95, 96,
97, 98, 99, 100]
(%i110) S:setify(S)
(%o110) {1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15, 16, 17, 18, 19, 20, 21, 22, 23, 24, 25, 26, 27, 28, 29, 30, 31, 32,
33, 34, 35, 36, 37, 38, 39, 40, 41, 42, 43, 44, 45, 46, 47, 48, 49, 50, 51, 52, 53, 54, 55, 56, 57, 58, 59, 60, 61, 62, 63, 64,
65, 66, 67, 68, 69, 70, 71, 72, 73, 74, 75, 76, 77, 78, 79, 80, 81, 82, 83, 84, 85, 86, 87, 88, 89, 90, 91, 92, 93, 94, 95, 96,
97, 98, 99, 100}
(%i111) m:cardinality(S)

```

```

(%o111) 100
(%i112) T4:subset(S,lambda([e],is(mod(e,4)=0)))
(%o112) {4, 8, 12, 16, 20, 24, 28, 32, 36, 40, 44, 48, 52, 56, 60, 64, 68, 72, 76, 80, 84, 88, 92, 96, 100}
(%i113) T6:subset(S,lambda([e],is(mod(e,6)=0)))
(%o113) {6, 12, 18, 24, 30, 36, 42, 48, 54, 60, 66, 72, 78, 84, 90, 96}
(%i114) T4and6:intersection(T4,T6)
(%o114) {12, 24, 36, 48, 60, 72, 84, 96}
(%i115) T4or6:union(T4,T6)
(%o115) {4, 6, 8, 12, 16, 18, 20, 24, 28, 30, 32, 36, 40, 42, 44, 48, 52, 54, 56, 60, 64, 66, 68, 72, 76, 78, 80, 84, 88, 90, 92, 96, 100}
(%i116) g[T4or6]:cardinality(T4or6)
(%o117) 33
(%i118) W[T4or6]:g[T4or6]/m
(%o118)  $\frac{33}{100}$ 
(%i119) g[T4]:cardinality(T4)
(%o119) 25
(%i120) W[T4]:g[T4]/m
(%o120)  $\frac{1}{4}$ 
(%i121) g[T6]:cardinality(T6)
(%o121) 16
(%i122) W[T6]:g[T6]/m
(%o122)  $\frac{4}{25}$ 
(%i123) g[T4and6]:cardinality(T4and6)
(%o123) 8
(%i124) W[T4and6]:g[T4and6]/m
(%o124)  $\frac{2}{25}$ 
(%i125) W:W[T4]+W[T6]-W[T4and6]
(%o125)  $\frac{33}{100}$ 
(%i126)

```

## Wahrscheinlichkeitsverteilungen

### Binomialverteilung

**Beispiel 27.** Gegeben sind  $n$  und  $p$  einer Binomialverteilung. Man berechne die Binomialverteilung und stelle diese graphisch dar.

```

(%i1) n:10
(%o1) 10
(%i2) p:0.1
(%o2) 0.1
(%i3) W(k):=binomial(n,k)*p**k*(1-p)**(n-k)
(%o3)  $W(k) := \binom{n}{k} p^k (1-p)^{n-k}$ 
(%i4) kumW(k):=sum(W(i),i,0,k)
(%o4)  $kumW(k) := \sum(W(i), i, 0, k)$ 
(%i5) for j:0 thru n do print(j, " ",floor(W(j)*1000+0.5)/1000.0,"
",floor(kumW(j)*1000+0.5)/1000.0)
0 0.349 0.349
1 0.387 0.736
2 0.194 0.93
3 0.057 0.987
4 0.011 0.998
5 0.001 1.0
6 0.0 1.0
7 0.0 1.0
8 0.0 1.0
9 0.0 1.0
10 0.0 1.0
(%o7) done
(%i8) plot2d([W(x)], [x,0,n])$
(%i9)

```





Abbildung 4. Binomialverteilung

**Aufgabe 8.** Erstelle die Grafik einer Binomialverteilung mit  $n=10$  und  $p=1/2$

Lösung:



**Beispiel 28.** Man bestimme für gegebene Werte von  $n$  und  $p$

- die Dichtefunktion der Binomialverteilung
- die Binomialverteilung
- der Erwartungswert
- die Varianz
- die Streuung

```
(%i1) load(distrib)$
(%i2) n:5
(%o2) 5
(%i3) p:0.1
(%o3) 0.1
(%i4) Dichte(x):=pdf_binomial(x,n,p)
(%o4) Dichte(x):= pdf binomial(x, n, p)
(%i5) D:makeList(Dichte(x),x,0,n)
(%o6) [0.59049, 0.32805, 0.0729, 0.0081, 4.500000000000002 × 10-4, 1.0000000000000006 × 10-5]
(%i7) Verteilung(x):=cdf_binomial(x,n,p)
```

```

(%o9) Verteilung(x) := cdf binomial(x, n, p)
(%i10) V:makelist(Verteilung(x),x,0,n)
(%o10) [0.59049, 0.91854, 0.99144, 0.99954, 0.99999, 1]
(%i11) E:n*p;V:n*p*(1-p);S:sqrt(V)
(%o11) 0.5
(%o12) 0.45
(%o13) 0.67082039324994
(%i14) E:mean_binomial(n,p)
(%o14) 0.5
(%i15) V:var_binomial(n,p)
(%o15) 0.45
(%i16) S:std_binomial(n,p)
(%o16) 0.67082039324994
(%i17)

```

**Aufgabe 9.** Binomialverteilung: Man bestimme mit Hilfe des Unterprogrammes „distrib“ die Dichtefunktion, die Verteilungsfunktion, den Erwartungswert, die Varianz und die Streuung für folgende Werte:

Aufgabe	Umfang n	Wahrscheinlichkeit p
a)	10	0,2
b)	7	1/6
c)	8	0,05
d)	15	0,2

**Beispiel 29.** Ein Galtonbrett ist ein mechanisches Modell zur Erklärung der Binomialverteilung. Die grundlegende Wahrscheinlichkeit ist  $1/2$

```

(%i1) p:1/2
(%o1) 1/2
(%i2) W(n,k):=binomial(n,k)*p**k*(1-p)**(n-k)
(%o2) W(n,k):=binomial(n,k)*p**k*(1-p)**(n-k)
(%i3) f(n):=makelist(W(n,k),k,0,n)
(%o3) f(n):=makelist(W(n,k),k,0,n)
(%i4) nagelreihen:5
(%o5) 5
(%i6) for i:0 thru nagelreihen do print(f(i))

```

```

[1]
[1/2, 1/2]
[1/4, 1/2, 1/4]
[1/8, 3/8, 3/8, 1/8]

```

$$\begin{bmatrix} \frac{1}{16}, \frac{1}{4}, \frac{3}{8}, \frac{1}{4}, \frac{1}{16} \\ \frac{1}{32}, \frac{5}{32}, \frac{5}{16}, \frac{5}{16}, \frac{5}{32}, \frac{1}{32} \end{bmatrix}$$

(%o6) done

(%i7)

## Poissonverteilung

**Beispiel 30.** Eine Telefonzentrale erhält im Mittel zwei Gespräche je Minute. Wie groß ist die Wahrscheinlichkeit, dass in einer Minute  $k=0,1,2,3,4,5,6$  Anmeldungen eintreffen, wenn die Zahl der Anmeldungen poissonverteilt angenommen wird?

(%i1) m:2

(%o1) 2

(%i2) W(k):=m\*\*k/k!\*exp(-m)

(%o2)  $W(k) := \frac{m^k}{k!} \exp(-m)$

(%i3) p:makelist(floor(W(k)\*1000+0.5)/1000.0,k,0,6),numer

(%o3) [0.135, 0.271, 0.271, 0.18, 0.09, 0.036, 0.012]

(%i4)

**Beispiel 31.** Eine Schreibkraft macht im Durchschnitt bei 1000 Wörtern nur einen Tippfehler. Im Durchschnitt enthält eine Seite genau 250 Wörter. Wie groß ist die Wahrscheinlichkeit, dass eine einzelne Seite 0,1,2,3,... Fehler enthält, wenn eine Poissonverteilung angenommen wird?

(%i25) m:250/1000

(%o25)  $\frac{1}{4}$

(%i26) W(k):=m\*\*k/k!\*exp(-m)

(%o26)  $W(k) := \frac{m^k}{k!} \exp(-m)$

(%i27) p:makelist(W(k),k,0,3),numer

(%o28) [0.7788007830714, 0.19470019576785, 0.024337524470981, 0.0020281270392485]

(%i29)

**Beispiel 32.** Man bestimme für gegebene Werte von m

- die Dichtefunktion der Poissonverteilung
- die Poissonverteilung

(%i1) m:0.5

(%o1) 0.5

(%i2) load(distrib)\$


```

(%i3) Dichte(x):=pdf_poisson(x,m)
(%o3) Dichte(x):= pdf_poisson(x, m)
(%i4) Verteilung(x):=cdf_poisson(x,m)
(%o4) Verteilung(x):= cdf_poisson(x, m)
(%i5) D:makelist(Dichte(x),x,0,5)
(%o5) [0.60653065971263, 0.30326532985632, 0.075816332464079, 0.01263605541068, 0.001579506926335,
1.5795069263349829 × 10-4]
(%i6) V:makelist(Verteilung(x),x,0,5)
(%o6) [0.60653065971263, 0.90979598956895, 0.98561232203303, 0.99824837744371, 0.99982788437004,
0.99998583506268]
(%i7)

```

**Aufgabe 10.** Man vergleiche die graphische Darstellung einer Binomialverteilung mit  $n=10$  und  $p=0,1$  mit ihrer Poissonnäherung.

Lösung:



Die Übereinstimmung ist ziemlich gut!

## Normalverteilung

**Beispiel 33.** Die Berechnung der standardisierten Normalverteilung erfolgt mit Hilfe der Integralrechnung (Flächenberechnung)! Für einen bestimmten Wert  $x_1$  sollen Dichte und Wahrscheinlichkeit bestimmt werden.

```

(%i6) x1:0.5
(%o6) 0.5

```

```

(%i7) u(x):=1/sqrt(2*pi)*exp(-x**2/2)
(%o7)  $u(x) := \frac{1}{\sqrt{2\pi}} \exp\left(-\frac{x^2}{2}\right)$ 
(%i8) u(x1),numer
(%o9) 0.3520653267643
(%i10) f(x):=1/sqrt(2*pi)*integrate(exp(-t**2/2),t,minf,x)
(%o10)  $f(x) := \frac{1}{\sqrt{2\pi}} \int_{-\infty}^x \exp\left(-\frac{t^2}{2}\right) dt$ 
(%i11) f(x1),numer
(%o13) 0.69146245362854
(%i14)

```

Das Unterprogramm „distrib“ ermöglicht einen einfach Zugang zur Verwendung der Normalverteilung. Die Verwendung von Normalverteilungstabellen ist nicht mehr erforderlich.

**Beispiel 34.** In einem kleinen Krämerladen werden im Durchschnitt täglich 30 Brotwecken verkauft, die Streuung ist 5. Wie groß ist die Wahrscheinlichkeit einer Verknappung, wenn täglich 34 Wecken bestellt werden?

```

(%i1) m:30
(%o1) 30
(%i2) s:5
(%o2) 5
(%i3) load(distrib)$
(%i4) W_ueber_34:1-cdf_normal(34,m,s),numer
(%o6) 0.2118553985834
(%i7)

```

**Beispiel 35.** Man berechne die standariszierte Normalverteilung für  $x=0,5$  mit dem Unterprogramm „distrib“.

```

(%i1) load(distrib)$
(%i2) pdf_normal(0.5,0,1),numer
(%o3) 0.3520653267643
(%i4) cdf_normal(0.5,0,1),numer
(%o4) 0.69146246127401
(%i5)

```

Art der Verteilung	Dichtefunktion	Verteilungsfunktion
Binomialverteilung	pdf_binomial(x,n,p)	cdf_binomial(x,n,p)
Poissonverteilung	pdf_poisson(x,m)	cdf_poisson(x,m)
Normalverteilung	pdf_normal(x,m,s)	cdf_normal(x,m,s)

Tabelle 4. Verteilungsfunktionen aus „distrib“

## Tabellenverzeichnis

Durchschnitt und Vereinigung von Mengen . . . . .	5
Erzeugung von Mengen und Listen . . . . .	6
Mächtigkeit und Produktmenge . . . . .	7
Verteilungsfunktionen aus „distrib“ . . . . .	32

## Abbildungsverzeichnis

Durchschnitt von Mengen . . . . .	4
Vereinigung von Mengen . . . . .	5
Fundamentales Abzählprinzip . . . . .	8
Binomialverteilung . . . . .	27